


### Задача 15

( Задача 999. Сборник задач по теоретической механике. И.Н. Веселовский. М.: Государственное издательство технико-теоретической литературы, 1955. – 500 с. )

Однородные стержни АВ и ВС веса  $P$  и длины  $2a$  каждый скреплены шарниром В. Конец А закреплён неподвижным шарниром, а к концу С приложена горизонтальная сила численно равная  $\frac{P}{2}$ . Определить углы  $\alpha$  и  $\beta$  при равновесии.

Для решения задачи использовались следующие значения параметров:  
 $m=5 \text{ кг}$ ,  $a=0.5 \text{ м}$ ,




**Точный теоретический ответ**

$$\alpha = \arctg\left(\frac{1}{3}\right);$$

$$\beta = \arctg(1).$$

**Решение задачи в EULER**



Система состоит из трёх звеньев.

- Инерциальное звено (base). В проекте звено не отображается.
- Стержень АВ (body1). Звено отображается цилиндром solid1 и имеет массу  $m$ .
- Стержень ВС (body2). Звено отображается цилиндром solid2 и имеет массу  $m$ .

Связь инерциального звена со стержнем АВ и стержня АВ со стержнем ВС моделируется с помощью шарниров типа «пара вращения». Для задания силы используется силовой элемент force1 типа «сила по датчикам на одну точку». Для демпфирования колебаний системы при нахождении равновесного состояния используются силовые элементы force2, force3 типа «вращательно-поступательный демпфер между двумя точками» в точке А между звеньями base, body1 и в точке В между звеньями body1, body2.

Для нахождения углов  $\alpha$  и  $\beta$  в положении равновесия используется команда «Расчёт динамики движения» command1 с постоянным шагом интегрирования 0.01[s] на отрезке интегрирования 50 [s].

### **Результаты моделирования**

Относительное отличие полученного в EULER решения от теоретического

delta\_alpha = 0.000 000 014 [-];

delta\_beta = 0.000 000 007 [-].

### **Текст проекта в EULER**

```

scalar g=9.81[m/s2];
scalar m=5[kg];
scalar P=m*g;
scalar a=0.5[m];
scalar l=2*a;
scalar M=10[N m];
point point1=point( 0[m], 0[m], 0[m] );
point point2=point( 0[m], -l, 0[m] );
point point3=point( 0[m], -2*l, 0[m] );
scalar r_c=1[mm];
solid solid1=cylinder( point1, point2, r_c, mass = m );
solid solid2=cylinder( point2, point3, r_c, mass = m );
body base=body( color = RGB( 229, 229, 229 ) );
set ground = base;
body base < ( point1 );
body body1=body( color = RGB( 255, 153, 153 ) );
body body1 < ( solid1, point2 );
body body2=body( color = RGB( 255, 153, 102 ) );
body body2 < ( solid2, point3 );
joint joint1=rotational( base, body1, point1, projectZ );
joint joint2=rotational( body1, body2, point2, projectZ );
gravity gravity1=parallel( reverse( projectY ), g = g );
force force1=force( body2, point3, projectX, P/2, list( ), fixing = unlock: );
force force2=damperComplex( base, point1, body1, point1, 0 [ kg/ s ], 10 [ kg m2/ s rad ] );
force force3=damperComplex( body1, point2, body2, point2, 0 [ kg/ s ], 10 [ kg m2/ s rad ] );
vector vector1=vectorPP( point1, point2 );
vector vector2=vectorPP( point2, point3 );
sensor alpha=angleVV( reverse( projectY ), vector1, axisDirection = reverse( projectZ ) );
sensor beta=angleVV( reverse( projectY ), vector2, axisDirection = reverse( projectZ ) );
sensor alpha_theoretical=arctg(1/3)*1[ rad ];
sensor beta_theoretical=arctg(1)*1[rad];
sensor delta_alpha=abs((alpha-alpha_theoretical)/alpha_theoretical);
sensor delta_beta=abs((beta-beta_theoretical)/beta_theoretical);
command command1=constRK4( 5.0000e+001 [ s ], 1.00000e-002 [ s ] );

```

Λ//

Λ Список главных команд;  
set dynamics = command1;

Λ//

Λ Единицы измерения;  
set units = SI;